Problémamegoldás táblázatkezelővel

Ebben a fejezetben arra mutatunk be példákat, hogyan használhatjuk a táblázatkezelő programot más tantárgyakban, pl. földrajz-, matematika-, fizika- vagy nyelvtanórán.

Az Európai Unió országai

Első példánkban az Európai Unió adatait elemezzük a Központi Statisztikai Hivatal adatai alapján. A 2019-es adatokat a könyv weblapjáról elérhető *eu.xlsx* fájl tartalmazza. A táblázat oszlopai rendre a tagország nevét, az EU-ba való belépés évét, területét, népességét, GDP-jét, a születéskor várható élettartamot, a főváros nevét és a főváros lakóinak számát tartalmazza. A mértékegységeket a táblázat első sorában tüntettük fel.

	A	B	C	D	E	F	G	H	I	J
1	Ország	Belépés	Terület ezer km^2	Népesség millió fő	Néps fő/km^2	GDP mrd €	GDP egy főre (€)	Sz.v.ét. év	Főváros	Főv. nép. ezer fő
2	Ausztria	1995	83,88	8,88		398,68		81,80	Bécs	1 915,30
3	Belgium	alapító	30,53	11,50		473,09		81,70	Brüsszel	2 065,30
4	Bulgária	2007	110,37	6,98		60,68		75,00	Szófia	1 276,90
5	Ciprus	2004	9,25	0,88		21,94		82,90	Nicosia	269,50
6	Csehország	2004	78,87	10,67		220,20		79,10	Prága	1 298,80
7	Dánia	1973	42,92	5,81		310,94		81,00	Koppenhága	1 333,90

▶ Az Európai Unió tagországainak adatai a KSH alapján (Forrás: https://www.ksh.hu/stadat_eves_7)

Határozzuk meg az *E* oszlop celláiban, hogy mekkora az egyes országok népsűrűsége! A népsűrűséget a népesség és a terület hányadosa adja, ez például Ausztria esetén *D2/C2* lenne. Mivel a népesség millió főben, a terület pedig ezer km^2-ben van megadva, a hányadost még szorozni kell 1000-rel is, így a `=D2/C2*1000` képlet került az *E2*-es cellába. Hasonló módon a *G2*-es cellában az egy főre jutó GDP-t €-ban kifejezve az `=F2/D2*1000` képlettel kapjuk. (Érdemes megvizsgálnunk ezeket az értékeket: jelentős eltéréseket találunk az egyes tagországok között.)

Az *L2:N17* tartományban további kérdésekre keresünk választ. Például az *M2:M4* tartomány celláiban meghatározzuk az unió területét és népességét, ebből pedig kiszámoljuk a népsűrűségét:

```
M2: =SZUM(C2:C28)
M3: =SZUM(D2:D28)
M4: =M3/M2*1000
```

	L	M	N
1			
2	EU területe	4 131,93	ezer km^2
3	EU népessége	447,27	millió fő
4	EU átlagos népsűrűsége	108,25	fő/km^2
5			
6	Legnagyobb GDP	3 435,21	mrd €
7	GDP-je a legnagyobb	Németország	
8			
9	Magyarországnál nagyobb a…		
10	területe	10	
11	területe és népessége	8	
12	területe, népessége és GDP-je	8	
13			
14	1 főre jutó GDP		
15	teljes EU	31 139,07 €	
16	alapítók	37 493,69 €	
17	2000 után belépők	14 601,41 €	

▶ További kérdések és a válaszok

Melyik ország GDP-je a legnagyobb az Unióban? Határozzuk meg először a legnagyobb GDP értékét! A legnagyobb GDP-t az *M6*-os cellában a `=MAX(F2:F28)` képlet adja. Azt, hogy ez melyik országhoz tartozik, meghatározhatjuk például az *INDEX…HOL.VAN* függvényekkel, így az *M7*-es cellában a következő képlet szerepel:

```
=INDEX(A2:A28;HOL.VAN(M6;F2:F28;0))
```

Vajon hol helyezkedik el Magyarország az unió 27 tagországa között? Hány nálunk „nagyobb" található? Pontosabban hány olyan ország van, melynek hazánkénál nagyobb a területe; nagyobb a területe és a népessége, illetve nagyobb a területe, a népessége és a nemzeti jövedelme is? A három kérdésre a válaszokat például a *DARABTELI* és *DARABHATÖBB* függvénnyel kapjuk az *M10*-es, *M11*-es és *M12*-es cellákban:

```
=DARABTELI(C2:C28;">"&C19)
=DARABHATÖBB(C2:C28;">"&C19;D2:D28;">"&D19)
=DARABHATÖBB(C2:C28;”>”&C19;D2:D28;”>”&D19; F2:F28;”>”&F19)
```

Az Európai Unió 1 főre jutó GDP-jét – a népsűrűség számításához hasonlóan határozhatjuk meg az *M15*-ös cellában:

```
=SZUM(F2:F28)/SZUM(D2:D28)*1000.
```

Kicsit összetettebb a feladat, ha szeretnénk meghatározni az egy főre jutó GDP-t az *alapító* országok esetén, illetve a *2000 után belépő* országokban külön-külön is. Ezúttal feltételes összegzést kell végeznünk, amit például a *SZUMHATÖBB* függvény segítségével készíthetünk, így a két képlet az *M16*-os és *M17*-es cellákban:

```
=SZUMHATÖBB(F2:F28;B2:B28;"alapító")/
              SZUMHATÖBB(D2:D28;B2:B28;"alapító")*1000
=SZUMHATÖBB(F2:F28;B2:B28;">"&2000)/
              SZUMHATÖBB(D2:D28;B2:B28;">"&2000)*1000
```

Feladatok

1. Milyen következtetéseket vonhatunk le a kapott adatokból? Hazánk az EU többi országához képest nagy vagy kicsi? Jóval fejlettebbek-e az alapítók, mint a 2000 után belépők?
2. Határozzuk meg az *M19*-es cellában, hogy melyik ország fővárosa a legnépesebb!
3. Képlet alkalmazásával tegyünk a *K* oszlopba egy „+" jelet, ha az adott országban az egy főre jutó GDP az uniós átlagnál nagyobb, egyébként pedig egy „-" jelet!
4. Határozzuk meg képlettel az *M21*-es cellában, hogy hány olyan ország van, amelyben az egy főre jutó GDP nagyobb az uniós átlagnál! Hány ember él ezekben az országokban? A választ az *M22*-es cellában jelenítsük meg!
5. Vajon van-e kapcsolat az egy főre jutó GDP és a születéskor várható élettartam között? Hogyan tudnánk erre a kérdésre a táblázatkezelő eszközeinek felhasználásával választ adni?

Egyenletek megoldása

Másodfokú egyenlet megoldása

Mint matematikaórán tanultuk, az $ax^2 + bx + c = 0$ ($a \neq 0$) másodfokú egyenlet megoldása a $D = b^2 - 4ac$ diszkriminánstól függ. Ha $D < 0$, akkor nincs valós megoldás, ha $D = 0$, akkor egy valós megoldás van, ha pedig D > 0, akkor kettő. Ez utóbbi esetben a két valós megoldást az $x_{1,2} = \frac{-b \pm \sqrt{b^2-4ac}}{2a}$ megoldóképlet adja. Először ezt a megoldóképletet használjuk fel az egyenlet megoldására a táblázatkezelő program segítségével.

Hozzuk létre a táblázat munkalapján az alábbi ábrán látható feliratokat az 1. sorban, továbbá az *A* és *D* oszlopokban!

Ha feltételezzük, hogy a felhasználó olyan együtthatókat választott, amelyek két valós megoldást adnak, akkor például az *E2* és *E3* cellákba kerülő képletek:

E2: `=B3*B3-4*B2*B4`, *E3:* `=(-B3+GYÖK(E2))/(2*B2)`

Hogyan tudnánk a képleteket úgy továbbfejleszteni, hogy azok akkor is helyesen működjenek, ha a diszkrimináns negatív? Mivel ebben az esetben a *GYÖK()* függvény hibát jelez, lehetőségünk van a **HAHIBA**(*kif1, kif2)* hibakezelő függvény alkalmazására. Ha a *kif1* kifejezés hibátlanul működik, akkor ez a függvény annak eredményét írja a cellába, egyébként pedig a *kif2* értékét. Így az *E2*-es cellába kerülő képlet:

```
=HAHIBA((-B3+GYÖK(E2))/(2*B2);"Nincs valós megoldás!")
```

	A	B	C	D	E	F
1	**Az $ax^2+bx+c=0$ alakú egyenlet megoldása**					
2	a=	**3**		D=	**76**	
3	b=	**4**		x_1=	**0,7863**	
4	c=	**-5**		x_2=	**-2,1196**	

	A	B	C	D	E	F
1	**Az $ax^2+bx+c=0$ alakú egyenlet megoldása**					
2	a=	**3**		D=	**-44**	
3	b=	**4**		x_1=	**Nincs valós megoldás!**	
4	c=	**5**		x_2=	**Nincs valós megoldás!**	

▶ Másodfokú egyenlet megoldása

Egyenletek grafikus megoldása

Matematikaórán elhangzott, hogy algebrai úton legfeljebb negyedfokú egyenletek oldhatók meg, ugyanakkor már a harmadfokú egyenletek megoldóképletének alkalmazása helyett is gyakran gyorsabbak a közelítő megoldások.

Ebben a példában a $-0{,}5x^3 + 0{,}8x^2 - 0{,}6x + 1 = 0$ harmadfokú egyenletet fogjuk megoldani. Értéktáblázattal határozzuk meg az $f(x) = -0{,}5x^3 + 0{,}8x^2 - 0{,}6x + 1$ függvény értékeit –1 és +2 között 0,2 tizedenként, majd ábrázoljuk az adatokat pontdiagramon!

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1	**x**	-1,00	-0,80	-0,60	-0,40	-0,20	0,00	0,20	0,40	0,60	0,80	1,00	1,20	1,40	1,60	1,80	2,00
2	**f(x)**	2,90	2,25	1,76	1,40	1,16	1,00	0,91	0,86	0,82	0,78	0,70	0,57	0,36	0,04	-0,40	-1,00

▶ Az $f(x) = -0{,}5x^3 + 0{,}8x^2 - 0{,}6x + 1$ függvény értékei –1 és 2 között 0,2-es lépésközzel, 2 tizedesjegy pontossággal

Az ábrán a *B2*-es cellába a képlet került:

```
=-0,5*B1*B1*B1+0,8*B1*B1-0,6*B1+1
```

Az értéktábla alapján úgy tűnik, hogy az egyenlet gyöke 1,6 és 1,8 közé esik, erről a grafikonról is meggyőződhetünk. Ha a grafikonon csökkentjük a tengelyeken a lépésközt, akkor pontosabb eredményt is leolvashatunk: a gyök 1,62 körül van.



▶ A függvény grafikonja



▶ Csökkentettük a tengelyen a lépésközt (Microsoft Excel)

Egyenletek megoldása célértékkereséssel

A gyökre becslést kaphatunk a táblázatkezelő program célértékkeresés funkciójával is. Ezt például az *Adatok > Lehetőségelemzés > Célértékkeresés (illetve az Eszközök > Célértékkeresés)* menüponttal érhetjük el.

Használatához adjuk meg a vizsgált képletet tartalmazó cellát *(B2)*, a várt eredményt (0), illetve azt a cellát, amely a változót tartalmazza *(B1)*. A táblázatkezelő program a változót tartalmazó cella módszeres változtatásával megpróbálja elérni a kért eredményt.

Esetünkben az egyenlet gyöke x = 1,6209 körül van, ekkor a függvényérték már csak kb. 0,00003-del tér el zérustól.



	A	B
1	x	1,620886178
2	f(x)	0,000031458

▶ Célértékkeresés

Feladatok

1. Bővítsük tovább a másodfokú egyenletet megoldó munkalap képleteit úgy, hogy az jelezze azt is, ha csak egyetlen megoldás van!
2. Ábrázoljuk közös koordinátarendszerben a következő függvényeket! Milyen kapcsolatot láthatunk az egyes függvények grafikonjai között?
   $f(x) = x^2$ $g(x) = -x^2$ $h(x) = (x-2)^2$ $i(x) = x^2 - 2$ $j(x) = -(x-2)^2 - 2$
3. Keressük meg célértékkereséssel a következő egyenletek gyökeit:
   a) $x^3 - 3x^2 + x - 3 = 0$
   b) $x^4 - x^3 - 7x^2 + x + 6 = 0$

Ferde hajítás

A ferde hajítás a fizika egyik érdekes és rendkívül gyakorlatias problémája: milyen pályán mozog egy elrúgott focilabda vagy egy kilőtt ágyúgolyó? Feladatunk az, hogy meghatározzuk a ferdén elhajított test *x* és *y* koordinátáit a mozgás során. Készítsük el a táblázatot a minta és a leírás alapján!

Kezdetben legyen a test az origóban ($x = 0$ és $y = 0$), és hajítsuk el úgy, hogy kezdősebessége vízszintesen 6 m/s, függőlegesen pedig 8 m/s legyen! (Pitagorasz tételével könnyen ellenőrizhetjük, hogy a test kezdősebessége 10 m/s.) Ezeket az adatokat táblázatunk *A5:E6* tartományának celláiban találjuk.



▸ Az *x* és *y* komponenst külön vizsgáljuk

A továbbiakban választunk egy viszonylag rövid Δt időközt (például $\Delta t = 0{,}1$ s), és felbontjuk a mozgás időtartamát Δt időegységekre. Minden időegység végén megadjuk a test helyének és sebességének vízszintes és függőleges komponensét az időegység elején kapott értékekből a fizika szabályainak alkalmazásával. Az időegység értékét táblázatunk *B2*-es cellájában rögzítettük.

Ha (egyelőre) eltekintünk a légellenállástól, a test vízszintes sebességkomponense nem változik, vagyis az első időegység végén megegyezik az időegység elején kapott értékkel, így a *B7*-es cella tartalma a

```
=B6
```

képlet lesz.

A test vízszintes irányú elmozdulásának meghatározása – gondolva arra, hogy később figyelembe szeretnénk venni a közegellenállást is – a következőképpen történik. Először meghatározzuk a test átlagsebességét az adott időegységre vonatkozóan: (*B6*+*B7*)/2. Majd ebből a $\Delta s = v_{\text{átl}}\Delta t$ képlet alkalmazásával kiszámoljuk a vízszintes irányú elmozdulást: (*B6*+*B7*)/2*\$*B*\$2, és ezt hozzáadjuk az időegység elején kapott értékhez. Így a *D7*-es cellába a következő képlet kerül:

```
=D6+(B6+B7)/2*$B$2
```

	A	B	C	D	E
1	g=	10	m/s²		
2	Δt=	0,1	s		
3	k=				
4					
5	**Idő**	**v_x**	**v_y**	**x**	**y**
6	0,0	6,00	8,00	0,00	0,00
7	0,1	6,00	7,00	0,60	0,75
8	0,2	6,00	6,00	1,20	1,40
9	0,3	6,00	5,00	1,80	1,95
10	0,4	6,00	4,00	2,40	2,40
11	0,5	6,00	3,00	3,00	2,75
12	0,6	6,00	2,00	3,60	3,00
13	0,7	6,00	1,00	4,20	3,15
14	0,8	6,00	0,00	4,80	3,20
15	0,9	6,00	-1,00	5,40	3,15
16	1,0	6,00	-2,00	6,00	3,00

▸ A ferde hajítást végző test pályájának és sebességének x és y irányú komponensei

A függőleges komponens vizsgálata (ha eltekintünk a légellenállástól) csak annyival bonyolultabb, hogy figyelembe kell venni a Föld tömegvonzását is.

A nehézségi gyorsulás *(g)* értékét a *B1*-es cellában rögzítjük (így később a számításokat más bolygókra is elvégezhetjük).

A test függőleges sebessége Δt időegység alatt $g\Delta t$-vel csökken, így a *C7*-es cellába a következő képlet került:

```
=C6-$B$1*$B$2
```

Végül a test y koordinátáját a korábban alkalmazott $\Delta s = v_{\text{átl}}\Delta t$ képlet alkalmazásával az *E7*-es cellában a

```
=E6+(C6+C7)/2*$B$2
```

képlettel kapjuk.

Az eredményt látványosabbá tehetjük, ha a mozgás során a koordinátákat pontdiagramon ábrázoljuk.



► A test koordinátáinak megjelenítése pontdiagramon

Feladatok

1. A képletek kialakítása során nem vettük figyelembe, hogy ha a test visszaér a felszínre ($x = 0$), akkor nem tud „lejjebb esni”. Hogyan kell módosítanunk ennek figyelembevételéhez a képleteket?
2. Hogyan mozogna a ferdén elhajított test a Holdon (g = 1,64 m/s^2), a Napon (g = 274 m/s^2), illetve egy olyan helyen, ahol nincs nagy tömegű test a közelben (g = 0 m/s^2)?
3. A légellenállás kis sebességek esetén a sebességgel egyenesen arányos, az arányossági tényező a test alakjától és felületétől függ. Példánkban ezt az arányossági tényezőt a *B3*-as cella tartalmazza, legyen értéke például 0,4. Hogyan változik a test pályája, ha a légellenállást is figyelembe vesszük?

Betűk gyakorisága

A betűk gyakorisága a magyar nyelvben

Először meghatározzuk az egyes betűk gyakoriságát egy magyar nyelvű szövegben. Töltsük le a könyv weboldaláról a *legyjo.txt* szöveges állományt, amely Móricz Zsigmond Légy jó mindhalálig! című regényének ékezetmentes változatát tartalmazza szóközök, számok és írásjelek nélkül.

Másoljuk be az *A2*-es cellától kezdve a szöveget a táblázatkezelő munkalapjára! A többi feliratot a minta szerint készítsük el! (A fájlban a 405 589 karakter mindegyike külön sorban szerepel.)



▶ A magyar nyelvben a két leggyakoribb betű az A és az E

	A	B	C	D	E	F	G
1	**nyílt**	**titk**		**betű**	**db**	**k**	**kulcs**
2	M	N		A	49960	123	T
3	O	C		B	7631	19	A
4	R	U		C	3854	10	B
5	I	P		D	9072	22	L
6	C	B		E	50786	125	Z
7	Z	J		F	3789	9	K
8	Z	J		G	15357	38	E
9	S	V		H	6923	17	O
10	I	P		I	19363	48	P
11	G	E		J	4736	12	R
12	M	N		K	18173	45	G
13	O	C		L	22754	56	M
14	N	W		M	16213	40	N
15	D	L		N	26123	64	W
16	L	M		O	29220	72	C
17	E	Z		P	3393	8	I
18	G	E		Q	7	0	Q
19	Y	F		R	16107	40	U
20	J	R		S	23872	59	V
21	O	C		T	36219	89	S
22	M	N		U	7458	18	X
23	I	P		V	8018	20	H
24	N	W		W	4	0	D
25	D	L		X	7	0	Y
26	H	O		Y	10620	26	F
27	A	T		Z	15930	39	J

▶ A betűk száma

Vegyük fel az ábécé betűit ékezetek nélkül a *D2:D27* tartományban, majd határozzuk meg azok gyakoriságát! A *DARABTELI* függvényt alkalmazva az *E2*-es cellában szereplő képlet: `=DARABTELI($A$1:$A$405590;D2)`.

A mintát könnyebb összehasonlítani egy másik szöveggel, ha a betűk darabszáma helyett azt adjuk meg, hogy az adott betű átlagosan hányszor fordul elő a szöveg 1000 karakterében. Például az *F2*-es cellában: `=E2/SZUM($E$2:$E$27)*1000`.

Az eredményt szemléltessük oszlopdiagramon!

Titkosítás más betűk behelyettesítésével

A titkosírások egyik régi technikája, amikor az ábécé egyes betűit az ábécé más betűivel helyettesítik. Egy ilyen titkosításhoz tartozó kódtáblát látunk az ábrán a *D2:D27;G2:G27* nem összefüggő tartományban.

A kódtábla alkalmazása a *B* oszlopban történhet például az *INDEX...HOL.VAN* függvénypárossal, így a *B2*-es cellába a következő másolható képlet írható:

```
=INDEX($G$2:$G$27; HOL.VAN(A2;$D$2:$D$27;0))
```

A titkosított szöveget a *B* oszlopból a vágólapon át könnyen áthelyezhetjük egy editorba (pl. a Notepad++), ahol a *Csere* funkcióval a sortörések eltávolíthatók.

A betűhelyettesítéssel kapott titkos szöveg feltörése

Vajon mennyire megbízható ez az eljárás? Matematikailag könnyen látható, hogy egy 26 betűs ábécé esetén az *A* betű 26-féleképpen helyettesíthető, a *B* 25-féleképpen, és így tovább, vagyis a lehetséges kódtáblák száma:

$26! \cong 400\,000\,000\,000\,000\,000\,000\,000\,000.$

Valamennyi eset kipróbálása a középkorban lehetetlen volt. Azonban már a 11. században felismerték, hogy az így titkosított szöveg a betűgyakoriság vizsgálatával megfejthető.

Töltsük be a könyv weblapjáról letölthető *titkos.txt* fájlt, és tartalmát illesszük a munkalap *A* oszlopába! Az előző példának megfelelően határozzuk meg, hogy az egyes betűk átlagosan hányszor fordulnak elő a szöveg 1000 karakterében! Az ábrán látható, hogy a *G* oszlopba felvettük az előző mintánál kapott megfelelő adatokat is.

Hasonlítsuk össze a mostani szövegben, illetve az előző példában végzett számításokat, vagyis az *F* és *G* oszlopok adatait! Az előző példa oszlopdiagramjáról leolvasható, hogy egy magyar nyelvű szövegben a leggyakoribb betű az *A* és az *E*. Így kialakulhat az a sejtésünk, hogy a titkos szövegben az *A* és *E* betű megfelelője az *O* és az *S* betű lehet. Vezessük át ezt a *B* oszlopba!

Tovább vizsgálva a szöveget, feltűnhet, hogy a vélhetőleg *E* betűvel kezdődő szövegrészek esetén a titkos szövegben az első három betű gyakran az *SHQ* betűhármas. Mivel a magyarban gyakori az *EGY* határozatlan névelő, feltételezhetjük, hogy az *EGY* betűcsoportnak az *SHQ*, vagyis a *G*-nek a *H*, míg a *Q*-nak az *S* betű felel meg… Az eljárást lépésről lépésre folytatva lassan visszafejthetjük a kódoláskor használt betűcseréket, és így megjelenik a *B* oszlopban a titkosított szöveg, Molnár Ferenc Pál utcai fiúk című regénye…

	A	B	C	D	E	F	G
1	L	I			db_{sz}	k_{sz}	k_m
2	V	H		A	4691	22	123
3	O	A		B	4368	20	19
4	N	R		C	12860	60	10
5	M	O		D	0	0	22
6	W	M		E	19943	93	125
7	C	N		F	11470	53	9
8	S	E		G	12382	57	38
9	H	G		H	7739	36	17
10	Q	Y		I	2238	10	48
11	S	E		J	73	0	12
12	A	D		K	4437	21	45
13	S	E		L	8814	41	56
14	H	G		M	16072	75	40
15	Q	Y		N	8384	39	64
16	F	K		O	27369	127	72
17	M	O		P	8214	38	8
18	N	R		Q	4665	22	0
19	S	E		R	2471	11	40
20	U	P		S	27453	127	59
21	U	P		T	2128	10	89
22	O	A		U	2040	9	18
23	K	B		V	3677	17	20
24	K	B		W	8102	38	0
25	O	A		X	11845	55	0
26	C	N		Y	3974	18	26
27	O	A		Z	4	0	39

▶ Titkos szöveg visszafejtése

Feladatok

1. Egy érdekes titkosítási eljárás, amikor az *A*-t *Z*-vel, *B*-t *Y*-nal… cseréljük és fordítva. Írjunk be egy közmondást karakterenként (ékezetek, szóközök és egyéb írásjelek nélkül) az *A* oszlopba, és rejtjelezzük ezzel az eljárással!
2. Töltsük le Mikszáth Kálmán Gavallérok című regényének szövegét a Magyar Elektronikus Könyvtárból! Távolítsuk el a szóközöket, számokat és egyéb írásjeleket, valamint az ékezeteket, majd a betűket cseréljük le nagybetűre és tördeljük új sorba! Rejtjelezzük a kapott szöveget az 1. feladatban leírt eljárással!